CS3021 Computer Architecture II - Tutorial 2

Student # 16327446 - Brandon Dooley

1) As a prerequisite, my code does work, build and run correctly

2) **Maximum Stack Depth (in Frames) =** 4 (See diagram below)

r9
r8
rdx = [b->21]
rcx = [a->14]
ret
r9
r8
rdx = [b->14]
rcx = [a->21]
ret
r9
r8
rdx = [b->7]
rcx = [a->14]
ret
r9
r8
rdx = [b->0]
rcx = [a->7]
ret

3) Code

[t2.asm] - min

```
includelib legacy_stdio_definitions.lib
extrn printf:near
option casemap : none                           ; case sensitive

.data
public g
g QWORD 4                                                               ;DD 64 bits 8 bytes
fxp1 db 'qns', 0AH, 00H                                                 ; ASCII format string
fxp2 db 'a = %I64d b = %I64d c = %I64d d = %I64d e = %I64d sum = %I64d', 0AH, 00H   ; ASCII format string

.code

; function to calculate min(a, b, c)
;             a -> rcx
;             b -> rdx
;             c -> r8
;
; returns result in rax

public  min                 ; export function name

min:
    mov rax, rcx            ; v = a (rcx)
    cmp rdx, rax            ; if (b < v)
    jge min1                ;
    mov rax, rdx            ; v = b
min1:
    cmp r8, rax             ; if (c < v)
    jge min2                ;
    mov rax, r8             ;  v = c
min2:
    ret 0                   ;
```

[t2.asm] - p

```
public   p                  ; export function name

p:
    push r8                 ; preserve k (r8)
    mov r8, rdx             ; pass j to min call in r8
    mov rdx, rcx            ; pass i to min call in rdx
    mov rcx, g              ; pass g to min call in rcx

    sub rsp, 32             ; allocate shadow space
    call min                ; min(g, i, j)
    add rsp, 32             ; deallocate shadow shadowspace

    mov r8, r9              ; pass l to min call in r8
    pop rdx                 ; pop k(r8) to min call in rdx
    mov rcx, rax            ; pass min(g, i, j) to min call in rcx
    call min                ; rax = min(min(g, i, j), k, l)
    ret 0
```

[t2.asm] - gcd

```
66  public  gcd                 ; export function name
67
68  gcd:
69      cmp rdx, 0              ; if(b==0)
70      je gcd_retA             ; return a
71
72      mov rax, rcx            ; rax = a (dividend)
73      mov rcx, rdx            ; rcx = b (divisor)
74      cqo                     ; sign extend rax into rdx
75      idiv rcx                ; rdx = a % b
76      sub rsp, 32             ; allocate shadow space
77
78      call gcd                ; rax = gcd(b, (a%b)) -> [b -> rcx, (a%b) -> rdx]
79
80      add rsp, 32             ; deallocate shadow space
81      jmp gcd_done            ;
82
83  gcd_retA:
84      mov rax, rcx            ; rax = a
85
86  gcd_done:
87      ret 0                   ; return
88
```

[t2.asm] - q

```
101  public   q                    ; export function name
102
103  q:
104      push rbp
105      mov  rbp, rsp              ; extract rsp
106      push rbx                   ; push non-volatile register
107
108
109      lea rax,[rcx+rdx]          ; rax = a+b
110      add rax,r8                 ; rax += c
111      add rax,r9                 ; rax += d
112      add rax,[rbp+48]           ; rax += e
113      push rax                   ; preserve sum (rax)
114
115      push rax                   ; push sum to printf stack
116      push [rbp+48]              ; push e to printf stack
117      push r9                    ; push d to printf stack
118      mov  r9,  r8               ; r9 -> c
119      mov  r8,  rdx              ; r8 -> b
120      mov  rdx, rcx              ; rdx -> a
121      lea  rcx, fxp2             ; rcx -> ascii format string address
122      sub  rsp, 32               ; allocate shadowspace for printf
123
124      call printf
125
126      add  rsp, 32               ; de-allocate shadowspace
127      add  rsp, 24               ; pop 3 (8 x 3) params off stack
128      pop  rax                   ; pop preserved sum off stack
129      pop  rbx                   ; restore rbx
130
131      mov  rsp, rbp              ; restore rsp
132      pop  rbp                   ; restore rbp
133      ret 0
```

[t2.asm] - qns

```
147  public  qns
148
149  qns:
150    push rbp                ; preserve rbp
151    mov  rbp, rsp           ; extract rsp
152    push rbx                ; preserve ebx (non-volatile)
153
154    lea  rcx, fxp1          ; string address
155    sub rsp, 32             ; allocate shadow space
156
157    call printf
158
159    add rsp, 32             ; deallocate shadow space
160    xor rax, rax            ; rax = 0
161
162    pop  rbx                ; restore rbx (non-volatile)
163    mov  rsp, rbp           ; restore rsp
164    pop  rbp                ; restore rbp
165    ret                     ; return
```

[t2.h]

```
15  extern "C" _int64 g;                                          // g
16
17  extern "C" _int64 min(_int64, _int64, _int64);                // min
18  extern "C" _int64 p(_int64, _int64, _int64, _int64);          // p
19  extern "C" _int64 gcd(_int64, _int64);                        // gcd
20  extern "C" _int64 q(_int64, _int64, _int64, _int64, _int64); // q
21  extern "C" _int64 qns();                                      // qns
```

3) Build Success

3) Console Window (qns with Shadow Space)



3) Qns without Shadow Space



When qns is executed without shadow space being allocated it causes an exception to be thrown as a result of an access violation when executing the instruction at location 0x00007FFDD….